ODE's with Maple

We illustrate Maple's differential equation solving ability by looking at an example that gives us an explicit solution, y'= $\frac{2y}{t} + t^2 e^t$, y(1)=0. Notice how we enter the differential equation. Since y is the dependent variable and t the independent variable, we must always use y(t) instead of just y.

```
> restart;
> deq:=diff(y(t),t)=(2/t)*y(t)+t^2*exp(t);
```

$$deq := \frac{d}{dt} y(t) = \frac{2 y(t)}{t} + t^2 e^t$$

We use the command infolevel to determine the level that Maple shows us about its approach to solving a problem. The default level is 1, 5 the maximum level.

```
> infolevel[dsolve]:=5;
```

$$infolevel_{dsolve} := 5$$

We use dsolve to find the general solution to the differential equation.

```
> sol:=dsolve(deq,y(t));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
<- 1st order linear successful
```

$$sol := y(t) = \left(e^t + \_C1\right) t^2$$

We can use odetest to test the correctness of our solution. For a correct solution, 0 is returned, although sometimes we may need to do a simplification to get 0.

```
> odetest(%,deq);
```

$$0$$

We enter the initial condition and find the particular solution.

```
> init:=y(1)=0;
```

$$init := y(1) = 0$$

```
> deqsol:=dsolve({deq,init},y(t));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
<- 1st order linear successful
```

$$deqsol := y(t) = \left(e^t - e\right) t^2$$

```
> odetest(%,deq);
```

$$0$$

After setting infolevel back to 1, we load the DEtools package, which contains functions that help you work with differential equations.

```
> infolevel[dsolve]:=1;
```

$$infolevel_{dsolve} := 1$$

```
> with(DEtools);
```

[*AreSimilar*, *DEnormal*, *DEplot*, *DEplot3d*, *DEplot_polygon*, *DFactor*, *DFactorLCLM*, *DFactorsols*, *Dchangevar*, *FunctionDecomposition*, *GCRD*, *Gosper*, *Heunsols*, *Homomorphisms*, *IVPsol*, *IsHyperexponential*, *LCLM*, *MeijerGsols*, *MultiplicativeDecomposition*, *ODEInvariants*, *PDEchangecoords*, *PolynomialNormalForm*, *RationalCanonicalForm*, *ReduceHyperexp*, *RiemannPsols*, *Xchange*, *Xcommutator*, *Xgauge*, *Zeilberger*, *abelsol*, *adjoint*, *autonomous*, *bernoullisol*, *buildsol*, *buildsym*, *canoni*, *caseplot*, *casesplit*, *checkrank*, *chinisol*, *clairautsol*, *constcoeffsols*, *convertAlg*, *convertsys*, *dalembertsol*, *dcoeffs*, *de2diffop*, *dfieldplot*, *diff_table*, *diffop2de*, *dperiodic_sols*, *dpolyform*, *dsubs*, *eigenring*, *endomorphism_charpoly*, *equinv*, *eta_k*, *eulersols*, *exactsol*, *expsols*, *exterior_power*, *firint*, *firtest*, *formal_sol*, *gen_exp*, *generate_ic*, *genhomosol*, *gensys*, *hamilton_eqs*, *hypergeomsols*, *hyperode*, *indicialeq*, *infgen*, *initialdata*, *integrate_sols*, *intfactor*, *invariants*, *kovacicsols*, *leftdivision*, *liesol*, *line_int*, *linearsol*, *matrixDE*, *matrix_riccati*, *maxdimsystems*, *moser_reduce*, *muchange*, *mult*, *mutest*, *newton_polygon*, *normalG2*, *ode_int_y*, *ode_y1*, *odeadvisor*, *odepde*, *parametricsol*, *particularsol*, *phaseportrait*, *poincare*, *polysols*, *power_equivalent*, *rational_equivalent*, *ratsols*, *redode*, *reduceOrder*, *reduce_order*, *regular_parts*, *regularsp*, *remove_RootOf*, *riccati_system*, *riccatisol*, *rifread*, *rifsimp*, *rightdivision*, *rtaylor*, *separablesol*, *singularities*, *solve_group*, *super_reduce*, *symgen*, *symmetric_power*, *symmetric_product*, *symtest*, *transinv*, *translate*, *untranslate*, *varparam*, *zoom*]

Let's look at the direction field for our differential equation. Each arrow shows the direction of movement from a particular point in the plane and has slope $\frac{d}{dt} y(t)$.

```
> DEplot (deq,y(t),t=1..2,y=-10..10,color=magenta,arrows = medium);
```



We superimpose the graph of the solution curve on the direction field.

```
> DEplot (deq,y(t),t=1..2,y=-10..10,{[1,0]},linecolor=green,color=
  magenta,arrows = medium);
```



We do the following to find the solution at a particular value of *t*, such as t=1.5.

```
> q:=rhs(deqsol);
```

$$q := \left(e^t - e\right) t^2$$

```
> "y(1.5)"=evalf(eval(q,t=1.5));
```

$$\text{"y(1.5)"} = 3.967666297$$

We can also do this by changing the right hand side into a function.

```
> y:=unapply(q,t);
  "y(1.5)"=evalf(y(1.5));
  "y(2)"=evalf(y(2));;
```

$$y := t \rightarrow \left(e^t - e\right) t^2$$

$$\text{"y(1.5)"} = 3.967666297$$

$$\text{"y(2)"} = 18.68309709$$

Now let's look at the solutions corresponding to the initial conditions $y(1) = i$, $i = -5..5$. Let's superimpose these solutions on the direction field.

```
> y:='y';
```

$$y := y$$

```
> DEplot (deq,[y(t)],t=1..2,y=-10..10,{[1,-5],[1,-4],[1,-3],[1,-2],
  [1,-1],[1,0],[1,1],[1,2],[1,3],[1,4],[1,5]},linecolor=green,
```

```
color=magenta,arrows = medium);
```



Some differential equations cannot be solved by Maple. This is one reason why we have need of numerical methods. We look at $D(y)(t) = 1 + t\sin(t\,y(t))$.

```
> infolevel[dsolve]:=5;
```

$$infolevel_{dsolve} := 5$$

```
> deq2:=D(y)(t)=1+t*sin(t*y(t));
```

$$deq2 := D(y)(t) = 1 + t\sin(t\,y(t))$$

```
> deqsol2:=dsolve(deq2,y(t));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
trying separable
trying inverse linear
trying homogeneous types:
trying Chini
differential order: 1; looking for linear symmetries
trying exact
Looking for potential symmetries
trying inverse_Riccati
trying an equivalence to an Abel ODE
```

```
trying 1st order ODE linearizable_by_differentiation
--- Trying Lie symmetry methods, 1st order ---
 -> Computing symmetries using: way = 3
 -> Computing symmetries using: way = 4
 -> Computing symmetries using: way = 5
trying symmetry patterns for 1st order ODEs
-> trying a symmetry pattern of the form [F(x)*G(y), 0]
-> trying a symmetry pattern of the form [0, F(x)*G(y)]
-> trying symmetry patterns of the forms [F(x),G(y)] and [G(y),F(x)]
-> trying a symmetry pattern of the form [F(x),G(x)]
-> trying a symmetry pattern of the form [F(y),G(y)]
-> trying a symmetry pattern of the form [F(x)+G(y), 0]
-> trying a symmetry pattern of the form [0, F(x)+G(y)]
-> trying a symmetry pattern of the form [F(x),G(x)*y+H(x)]
-> trying a symmetry pattern of conformal type
```

$$deqsol2 :=$$

We see that Maple cannot find an exact solution. Let's look at the direction field.

```
> infolevel[dsolve]:=1;
```

$$infolevel_{dsolve} := 1$$

```
> DEplot (deq2,y(t),t=-5..5,y=-5..5,color=magenta,arrows = medium);
```



Kind of wild!